
Abstract. Contrary to several other families of lambda terms, no closed9

formula or generating function is known and none of the sophisticated10

techniques devised in analytic combinatorics can help with counting or11

generating the set of simply-typed closed lambda terms of a given size.12

Moreover, their asymptotic scarcity among the set of closed lambda13

terms makes counting them via brute force generation and type inference14

quickly intractable, with previous work showing counts for them only up15

to size 10.16

By taking advantage of the synergy between logic variables, unification17

with occurs check and efficient backtracking in today’s Prolog systems,18

we climb 4 orders of magnitude above previously known values by de-19

riving progressively faster Horn Clause programs that generate and/or20

count the set of closed simply-typed lambda terms of sizes up to 14. A21

similar count for closed simply-typed normal forms is also derived up to22

size 14.23

Keywords: logic programming transformations, type inference, combi-24

natorics of lambda terms, simply-typed lambda calculus, simply-typed nor-25

mal forms.26

1 Introduction27

Generation of lambda terms [1] has practical applications to testing compilers28

that rely on lambda calculus as an intermediate language, as well as in generation29

of random tests for user-level programs and data types. At the same time, several30

instances of lambda calculus are of significant theoretical interest given their31

correspondence with logic and proofs.32

Simply-typed lambda terms [2, 3] enjoy a number of nice properties, among33

which strong normalization (termination for all evaluation-orders), a cartesian34

closed category mapping and a set-theoretical semantics. More importantly, via35

the Curry-Howard correspondence lambda terms that are inhabitants of simple36

types can be seen as proofs for tautologies in minimal logic which, in turn, cor-37

respond to the types. Extended with a fix-point operator, simply-typed lambda38



terms can be used as the intermediate language for compiling Turing-complete39

functional languages. Random generation of simply-typed lambda terms can also40

help with automation of debugging compilers for functional programming lan-41

guages [4].42

Recent work on the combinatorics of lambda terms [5–8], relying on recursion43

equations, generating functions and techniques from analytic combinatorics [9]44

has provided counts for several families of lambda terms and clarified important45

properties like their asymptotic density. With the techniques provided by gener-46

ating functions [9], it was possible to separate the counting of the terms of a given47

size for several families of lambda terms from their more computation intensive48

generation, resulting in several additions (e.g., A220894, A224345, A114851) to49

The On-Line Encyclopedia of Integer Sequences, [10].50

On the other hand, the combinatorics of simply-typed lambda terms, given51

the absence of closed formulas, recurrence equations or grammar-based genera-52

tors, due to the intricate interaction between type inference and the applicative53

structure of lambda terms, has left important problems open, including the very54

basic one of counting the number of closed simply-typed lambda terms of a given55

size. At this point, obtaining counts for simply-typed lambda terms requires go-56

ing through the more computation-intensive generation process.57

As a fortunate synergy, Prolog’s sound unification of logic variables, back-58

tracking and and definite clause grammars has been shown to provide compact59

combinatorial generation algorithms for various families of lambda terms [11–14].60

For the case of simply-typed lambda terms, we have pushed (in the unpub-61

lished draft [15]) the counts in sequence A220471 of [10] to cover sizes 11 and 12,62

each requiring each about one magnitude of extra computation effort, simply by63

writing the generators in Prolog. In this paper we focus on going two more mag-64

nitudes higher, while also integrating the results described in [15]. Using similar65

techniques, we achieve the same, for the special case of simply-typed normal66

forms.67

The paper is organized as follows. Section 2 describes our representation of68

lambda terms and derives a generator for closed lambda terms. Section 3 defines69

generators for well-formed type formulas. Section 4 introduces a type inference70

algorithms and then derives, step by step, efficient generators for simply-typed71

lambda terms and simple types inhabited by terms of a given size. Section 572

defines generators for closed lambda terms in normal form and then replicates the73

derivation of efficient generator for simply-typed closed normal forms. Section 674

aggregates our experimental performance data and hints to future improvements.75

Section 7 discusses related work and section 8 concludes the paper.76

The paper is structured as a literate Prolog program. The code has been77

tested with SWI-Prolog 7.3.8 and YAP 6.3.4. It is also available as a separate78

file at http://www.cse.unt.edu/~tarau/research/2016/lgen.pro.79

2 Deriving a generator for lambda terms80

Lambda terms can be seen as Motzkin trees [16], also called unary-binary trees,81

labeled with lambda binders at their unary nodes and corresponding variables82

at the leaves. We will thus derive a generator for them from a generator for83

Motzkin trees.84

2.1 A canonical representation with logic variables85

We can represent lambda terms [1] in Prolog using the constructors a/2 for appli-86

cations, l/2 for lambda abstractions and v/1 for variable occurrences. Variables87

bound by the lambdas and their occurrences are represented as logic variables. As88

an example, the lambda term λa.(λb.(a (b b)) λc.(a (c c))) will be represented as89

l(A,a(l(B,a(v(A),a(v(B),v(B)))),l(C,a(v(A),a(v(C),v(C)))))). As vari-90

ables share a unique scope (the clause containing them), this representation as-91

sumes that distinct variables are used for distinct scopes induced by the lambda92

binders in terms occurring in a given Prolog clause.93

Lambda terms might contain free variables not associated to any binders.94

Such terms a called open. A closed term is such that each variable occurrence is95

associated to a binder.96

2.2 Generating Motzkin trees97

Motzkin-trees (also called binary-unary trees) have internal nodes of arities 1 or98

2. Thus they can be seen as a skeleton of lambda terms that ignores binders and99

variables and their leaves.100

The predicate motzkin/2 generates Motzkin trees with S internal and leaf101

nodes.102

motzkin(S,X):-motzkin(X,S,0).103

104

motzkin(v)-->[].105

motzkin(l(X))-->down,motzkin(X).106

motzkin(a(X,Y))-->down,motzkin(X),motzkin(Y).107

108

down(s(X),X).109

Motzkin-trees are counted by the sequence A001006 in [10]. If we replace110

the first clause of motzkin/2 with motzkinTree(u)-->[], we obtain binary-111

unary trees with L internal nodes, counted by the entry A006318 (Large Schröder112

Numbers) of [10].113

Note the use of the predicate down/2, that assumes natural numbers in unary114

notation, with n s/1 symbols wrapped around 0 to denote n ∈ N. As our com-115

binatorial generation algorithms will usually be tractable for values of n below116

15, the use of unary notation is comparable (an often slightly faster) than the117

call to arithmetic built-ins. Note also that this leads, after the DCG translation,118

to “pure” Prolog programs made exclusively of Horn Clauses.119

To more conveniently call these generators with the usual natural numbers120

we define the converter n2s as follows.121

n2s(0,0).122

n2s(N,s(X)):-N>0,N1 is N-1,n2s(N1,X).123

Example 1 Motzkin tress with 2 internal nodes.124

?- n2s(1,S),motzkin(S,T).125

S = s(0), T = l(v) ;126

S = s(0), T = a(v, v) .127

2.3 Generating closed lambda terms128

We derive a generator for closed lambda terms by adding logic variables as labels129

to their binder and variable nodes, while ensuring that the terms are closed, i.e.,130

that the function mapping variables to their binders is total.131

The predicate lambda/2 builds a list of logic variables as it generates binders.132

When generating a leaf variable, it picks “nondeterministically” one of the binders133

among the list of binders available, Vs. As in the case of Motzkin trees, the pred-134

icate down/2 controls the number of internal nodes.135

lambda(S,X):-lambda(X,[],S,0).136

137

lambda(v(V),Vs)-->{member(V,Vs)}.138

lambda(l(V,X),Vs)-->down,lambda(X,[V|Vs]).139

lambda(a(X,Y),Vs)-->down,lambda(X,Vs),lambda(Y,Vs).140

The sequence A220471 in [10] contains counts for lambda terms of increasing141

sizes, with size defined as the number of internal nodes.142

Example 2 Closed lambda terms with 2 internal nodes.143

?- lambda(s(s(0)),Term).144

Term = l(A, l(B, v(B))) ;145

Term = l(A, l(B, v(A))) ;146

Term = l(A, a(v(A), v(A))) .147

3 A visit to the other side: the language of types148

As a result of the Curry-Howard correspondence, the language of types is iso-149

morphic with that of minimal logic, with binary trees having variables at leaf150

positions and the implication operator (“->”) at internal nodes. We will rely151

on the right associativity of this operator in Prolog, that matches the standard152

notation in type theory.153

The predicate type skel/3 generates all binary trees with given number of154

internal nodes and labels their leaves with unique logic variables. It also collects155

the variables to a list returned as its third argument.156

type_skel(S,T,Vs):-type_skel(T,Vs,[],S,0).157

158

type_skel(V,[V|Vs],Vs)-->[].159

type_skel((X->Y),Vs1,Vs3)-->down,160

type_skel(X,Vs1,Vs2),161

type_skel(Y,Vs2,Vs3).162

Type skeletons are counted by the Catalan numbers (sequence A000108 in [10]).163

Example 3 All type skeletons for N=3.164

?- type_skel(s(s(s(0))),T,_).165

T = (A->B->C->D) ;166

T = (A-> (B->C)->D) ;167

T = ((A->B)->C->D) ;168

T = ((A->B->C)->D) ;169

T = (((A->B)->C)->D) .170

The next step toward generating the set of all type formulas is observing that171

logic variables define equivalence classes that can be used to generate partitions172

of the set of variables, simply by selectively unifying them.173

The predicate mpart of/2 takes a list of distinct logic variables and generates174

partitions-as-equivalence-relations by unifying them “nondeterministically”. It175

also collects the unique variables, defining the equivalence classes as a list given176

by its second argument.177

mpart_of([],[]).178

mpart_of([U|Xs],[U|Us]):-179

mcomplement_of(U,Xs,Rs),180

mpart_of(Rs,Us).181

To implement a set-partition generator, we will split a set repeatedly in182

subset+complement pairs with help from the predicate mcomplement of/2.183

mcomplement_of(_,[],[]).184

mcomplement_of(U,[X|Xs],NewZs):-185

mcomplement_of(U,Xs,Zs),186

mplace_element(U,X,Zs,NewZs).187

188

mplace_element(U,U,Zs,Zs).189

mplace_element(_,X,Zs,[X|Zs]).190

To generate set partitions of a set of variables of a given size, we build a list191

of fresh variables with the equivalent of Prolog’s length predicate working in192

unary notation len/2.193

partitions(S,Ps):-len(Ps,S),mpart_of(Ps,_).194

195

len([],0).196

len([_|Vs],s(L)):-len(Vs,L).197

The count of the resulting set-partitions (Bell numbers) corresponds to the198

entry A000110 in [10].199

Example 4 Set partitions of size 3 expressed as variable equalities.200

?- partitions(s(s(s(0))),P).201

P = [A, A, A] ;202

P = [A, B, A] ;203

P = [A, A, B] ;204

P = [A, B, B] ;205

P = [A, B, C].206

We can then define the language of formulas in minimal logic, among which207

tautologies will correspond to simple types, as being generated by the predicate208

maybe type/2.209

maybe_type(L,T,Us):-type_skel(L,T,Vs),mpart_of(Vs,Us).210

Example 5 Well-formed formulas of minimal logic (possibly types) of size 2.211

?- maybe_type(s(s(0)),T,_).212

T = (A->A->A) ;213

T = (A->B->A) ;214

T = (A->A->B) ;215

T = (A->B->B) ;216

T = (A->B->C) ;217

T = ((A->A)->A) ;218

T = ((A->B)->A) ;219

T = ((A->A)->B) ;220

T = ((A->B)->B) ;221

T = ((A->B)->C) .222

The sequence 2,10,75,728,8526,115764,1776060,30240210 counting these for-223

mulas corresponds to the the product of Catalan and Bell numbers.224

4 Merging the two worlds: generating simply-typable225

lambda terms226

One can observe that per-size counts of both the sets of lambda terms and their227

potential types are very fast growing. There’s an important difference, though,228

between computing the type of a given lambda term (if it exists) and computing229

an inhabitant of a type (if it exists). The first operation, called type inference is230

an efficient operation (linear in practice) while the second operation, called the231

inhabitation problem is P-space complete [17].232

This brings us to design a type inference algorithm that takes advantage of233

operations on logic variables.234

4.1 A type inference algorithm235

While in a functional language inferring types requires implementing unification236

with occurs-check, as shown for instance in [5], this operation is available in237

Prolog as a built-in, predicate, optimized, for instance, in SWI-Prolog [18], to238

proceed incrementally, only checking that no new cycles are introduced during239

the unification step as such.240

The predicate infer type/3 works by using logic variables as dictionaries241

associating terms to their types. Each logic variable is then bound to a term of242

the form X:T where X will be a component of a fresh copy of the term and T will243

be its type. Note that we create this new term as the original term’s variables244

end up loaded with chunks of the partial types created during the type inference245

process.246

As logic variable bindings propagate between binders and occurrences, this247

ensures that types are consistently inferred.248

infer_type((v(XT)),v(X),T):-unify_with_occurs_check(XT,X:T).249

infer_type(l((X:TX),A),l(X,NewA),(TX->TA)):-infer_type(A,NewA,TA).250

infer_type(a(A,B),a(X,Y),TY):-infer_type(A,X,(TX->TY)),infer_type(B,Y,TX).251

Example 6 illustrates typability of the term corresponding to the S combinator252

λx0.λx1.λx2.((x0 x2) (x1 x2))253

and untypabilty of the term corresponding to the Y combinator254

λx0.(λx1.(x0 (x1 x1)) λx2.(x0 (x2 x2))).255

?- infer_type(l(A,l(B,l(C,a(a(v(A),v(C)),a(v(B),v(C)))))),X,T),256

portray_clause((T:-X)),fail.257

(A->B->C)-> (A->B)->A->C :-258

l(D,l(F,l(E, a(a(v(D), v(E)), a(v(F), v(E)))))).259

260

?- infer_type(261

l(A,a(l(B,a(v(A),a(v(B),v(B)))),l(C,a(v(A),a(v(C),v(C)))))), X, T).262

false.263

By combining generation of lambda terms with type inference we have our264

first cut to an already surprisingly fast generator for simply-typable lambda265

terms, able to generate in a few hours counts for sizes 11 and 12 for sequence266

A220471 in [10].267

lamb_with_type(S,X,T):-lambda(S,XT),infer_type(XT,X,T).268

Example 7 Lambda terms of size up to 3 and their types.269

?- lamb_with_type(s(s(s(0))),Term,Type).270

Term = l(A, l(B, l(C, v(C)))), Type = (D->E->F->F) ;271

Term = l(A, l(B, l(C, v(B)))), Type = (D->E->F->E) ;272

Term = l(A, l(B, l(C, v(A)))), Type = (D->E->F->D) ;273

Term = l(A, l(B, a(v(B), v(A)))), Type = (C-> (C->D)->D) ;274

Term = l(A, l(B, a(v(A), v(B)))), Type = ((C->D)->C->D) ;275

Term = l(A, a(v(A), l(B, v(B)))), Type = (((C->C)->D)->D) ;276

Term = l(A, a(l(B, v(B)), v(A))), Type = (C->C) ;277

Term = l(A, a(l(B, v(A)), v(A))), Type = (C->C) ;278

Term = a(l(A, v(A)), l(B, v(B))), Type = (C->C).279

Note that, for instance, when one wants to select only terms having a given280

type, this is quite inefficient. Next, we will show how to combine size-bound term281

generation, testing for closed terms and type inference into a single predicate.282

This will enable more efficient querying for terms inhabiting a given type, as one283

would expect from Prolog’s multi-directional execution model, and more impor-284

tantly for our purposes, to climb two orders of magnitude higher for counting285

simply-typed terms of size 13 and 14.286

4.2 Combining term generation and type inference287

We need two changes to infer type to turn it into an efficient generator for288

simply-typed lambda terms. First, we need to add an argument to control the289

size of the terms and ensure termination, by calling down/2 for internal nodes.290

Second, we need to generate the mapping between binders and variables. We291

ensure this my borrowing the member/2-based mechanism used in the predicate292

lambda/4 generating closed lambda terms in subsection 2.3.293

The predicate typed lambda/3 does just that, with helper from DCG-expanded294

typed lambda/5.295

typed_lambda(S,X,T):-typed_lambda(_XT,X,T,[],S,0).296

297

typed_lambda(v(V:T),v(V),T,Vs)--> {298

member(V:T0,Vs),299

unify_with_occurs_check(T0,T)300

}.301

typed_lambda(l(X:TX,A),l(X,NewA),(TX->TY),Vs)-->down,302

typed_lambda(A,NewA,TY,[X:TX|Vs]).303

typed_lambda(a(A,B),a(NewA,NewB),TY,Vs)-->down,304

typed_lambda(A,NewA,(TX->TY),Vs),305

typed_lambda(B,NewB,TX,Vs).306

Like lambda/4, the predicate typed lambda/5 relies on Prolog’s DCG nota-307

tion to thread together the steps controlled by the predicate down. Note also308

the nondeterministic use of the built-in member/2 that enumerates values for309

variable:type pairs ranging over the list of available pairs Vs, as well as the310

use of unify with occurs check to ensure that unification of candidate types311

does not create cycles.312

We will discuss exact performance data later, but let’s note here that this313

operation brings down by an order of magnitude the computational effort to gen-314

erate simply-typed terms. As expected, the number of solutions is computed as315

the sequence A220471 in [10]. Interestingly, by interleaving generation of closed316

terms and type inference in the predicate typed lambda, the time to gener-317

ate all the well-typed terms is actually shorter than the time to generate all318

closed terms of the same size, e.g., 17.123 vs. 31.442 seconds for size 10 with319

SWI-Prolog. As, via the Curry-Howard isomorphism, closed simply typed terms320

correspond to proofs of tautologies in minimal logic, co-generation of terms and321

types corresponds to co-generation of tautologies and their proofs for proofs of322

given length.323

Example 8 A term of size 15 and its type.324

l(A,l(B,l(C,l(D,l(E,l(F,l(G,l(H,l(I,l(J,l(K,325

a(v(I),l(L,a(a(v(E),v(J)),v(J)))))))))))))))326

M->N->O->P-> (Q->Q->R)->S->T->U-> ((V->R)->W)->Q->X->W327

4.3 One more trim: generating inhabited types328

Let’s first observe that the actual lambda term does not need to be built, pro-329

vided that we mimic exactly the type inference operations that one would need330

to perform to ensure it is simply-typed. It is thus safe to remove the first ar-331

gument of typed lambda/5 as well as the building of the fresh copy performed332

in the second argument. To further simplify the code, we can also make the333

DCG-processing of the size computations explicit, in the last two arguments.334

This gives the predicate inhabited type/4 and then inhabited type/2,335

that generates all types having inhabitants of a given size, but omits the inhab-336

itants as such.337

inhabited_type(X,Vs,N,N):-338

member(V,Vs),339

unify_with_occurs_check(X,V).340

inhabited_type((X->Xs),Vs,s(N1),N2):-341

inhabited_type(Xs,[X|Vs],N1,N2).342

inhabited_type(Xs,Vs,s(N1),N3):-343

inhabited_type((X->Xs),Vs,N1,N2),344

inhabited_type(X,Vs,N2,N3).345

Clearly the multiset of generated types has the same count as the set of their346

inhabitants and this brings us a nice additional 1.5x speed-up.347

inhabited_type(S,T):-inhabited_type(T,[],S,0).348

One more step makes reaching counts for sizes 13 and 14 achievable: using a349

faster Prolog, with a similar unify with occurs check built-in, like YAP [19],350

with the last value computed on a MacAir in less than a day.351

Example 9 The sequence A220471 completed up to N=14352

first 10: 1,2,9,40,238,1564,11807,98529,904318,9006364353

354

11: 96,709,332355

12: 1,110,858,977356

357

13: 13,581,942,434358

14: 175,844,515,544359

5 Doing it once more: generating closed simply-typed360

normal forms361

We will devise similar methods for an important subclass of simply-typed lambda362

terms.363

5.1 Generating normal forms364

Normal forms are lambda terms that cannot be further reduced. A normal form365

should not be an application with a lambda as its left branch and, recursively,366

its subterms should also be normal forms. The predicate normal form/2 uses367

normal form/4 to define them inductively and generates all normal forms with368

S internal nodes.369

normal_form(S,T):-normal_form(T,[],S,0).370

371

normal_form(v(X),Vs)-->{member(X,Vs)}.372

normal_form(l(X,A),Vs)-->down,normal_form(A,[X|Vs]).373

normal_form(a(v(X),B),Vs)-->down,normal_form(v(X),Vs),normal_form(B,Vs).374

normal_form(a(a(X,Y),B),Vs)-->down,normal_form(a(X,Y),Vs),normal_form(B,Vs).375

Example 10 illustrates closed normal forms with 2 internal nodes.376

?- normal_form(s(s(0)),NF).377

NF = l(A, l(B, v(B))) ;378

NF = l(A, l(B, v(A))) ;379

NF = l(A, a(v(A), v(A))) .380

The number of solutions of our generator replicates entry A224345 in [10] that381

counts closed normal forms of various sizes.382

The predicate nf with type applies the type inference algorithm to the gen-383

erated normal forms of size S.384

nf_with_type(S,X,T):-normal_form(S,XT),infer_type(XT,X,T).385

5.2 Merging in type inference386

Like in the case of the set of simply-typed lambda terms, we can define the more387

efficient combined generator and type inferrer predicate typed nf/2.388

typed_nf(S,X,T):-typed_nf(_XT,X,T,[],S,0).389

It works by calling the DCG-expended typed nf/4 predicate, with the last390

two arguments enforcing the size constraints.391

typed_nf(v(V:T),v(V),T,Vs)--> {392

member(V:T0,Vs),393

unify_with_occurs_check(T0,T)394

}.395

typed_nf(l(X:TX,A),l(X,NewA),(TX->TY),Vs)-->down,396

typed_nf(A,NewA,TY,[X:TX|Vs]).397

typed_nf(a(v(A),B),a(NewA,NewB),TY,Vs)-->down,398

typed_nf(v(A),NewA,(TX->TY),Vs),399

typed_nf(B,NewB,TX,Vs).400

typed_nf(a(a(A1,A2),B),a(NewA,NewB),TY,Vs)-->down,401

typed_nf(a(A1,A2),NewA,(TX->TY),Vs),402

typed_nf(B,NewB,TX,Vs).403

Example 11 Simply-typed normal forms up to size 3.404

?- typed_nf(s(s(s(0))),Term,Type).405

Term = l(A, l(B, l(C, v(C)))),406

Type = (D->E->F->F) ;407

...408

Term = l(A, a(v(A), l(B, v(B)))),409

Type = (((C->C)->D)->D) .410

We are now able to efficiently generate counts for simply-typed normal forms411

of a given size.412

Example 12 Counts for closed simply-typed normal forms up to N=14.413

first 10: 1,2,6,23,108,618,4092,30413,252590,2297954414

415

11: 22,640,259416

12: 240,084,189417

13: 2,721,455,329418

14: 32,783,910,297419

6 Experimental data and discussion420

Size closed λ-terms gen, then infer gen + infer inhabitants typed normal form
1 15 19 16 9 19
2 44 59 50 28 47
3 166 261 188 113 127
4 810 1,517 864 553 429
5 4,905 10,930 4,652 3,112 1,814
6 35,372 92,661 28,878 19,955 9,247
7 294,697 895,154 202,526 143,431 55,219
8 2,776,174 9,647,495 1,586,880 1,146,116 377,745
9 29,103,799 114,273,833 13,722,618 10,073,400 2,896,982

10 335,379,436 1,471,373,474 129,817,948 96,626,916 24,556,921

Fig. 1. Number of logical inferences used by our generators, as counted by SWI-Prolog

Figure 1 gives the number of logical inferences as counted by SWI-Prolog.421

This is a good measure of computational effort except for counting operations like422

unify with occurs check as a single step, while it’s actual complexity depends423

on the size of the terms involved. Therefore, figure 2 gives actual timings for the424

same operations, starting at N=5 where they start to be meaningful.425

The “closed λ-terms” column gives logical inferences and timing for gen-426

erating all closed lambda terms of size given in column 1. The column “gen,427

then infer” gives data for the algorithm that first generates lambda terms and428

then infers their types. The column “gen + infer” gives performance data for429

the significantly faster for the algorithm that merges generation and type in-430

ference in the same predicate. The column “inhabitants” gives data for the431

case when actual inhabitants are omitted in the merged generation and type432

inference process. The column “typed normal form” shows result for the fast,433

merged generation and type inference for terms in normal form.434

Size closed λ-terms gen, then infer gen + infer inhabitants typed normal form
5 0.001 0.001 0.001 0.000 0.001
6 0.005 0.011 0.004 0.002 0.004
7 0.028 0.114 0.029 0.018 0.011
8 0.257 1.253 0.242 0.149 0.050
9 2.763 15.256 2.080 1.298 0.379

10 32.239 199.188 19.888 12.664 3.329

Fig. 2. Timings (in seconds) for our generators up to size 10 on a 2015 MacBook

As moving from a size to the next typically adds one order of magnitude of435

computational effort, computing values for N=15 and N=16 is reachable with our436

best algorithms for both simply typed terms and their normal form subset.437

An interesting open problem is if this can be pushed significantly farther. We438

have looked into term hash based indexing and tabling-based dynamic program-439

ming algorithms, using de Bruijn terms. Unfortunately as subterms of closed440

terms are not necessarily closed, even if de Bruijn terms can be used as ground441

keys, their associated types are incomplete and dependent on the context in442

which they are inferred.443

While it only offers a constant factor speed-up, parallel execution is a more444

promising possibility. However, given the small granularity of the generation445

and type inference process, the most useful parallel execution mechanism would446

simply split the task of combined generation and inference process into a number447

of disjoint sets, corresponding to the number of available processors. A way to448

do this, is by using unranking functions (bijections originating in N) to the sets449

of combinatorial objects involved, and then, for k processors, assign work on450

successive numbers belonging to the same equivalence class modulo k.451

We have not seen any obvious way to improve these results using constraint452

programing systems, partly because the key “inner loop” computation is unifi-453

cation with occurs check with computations ranging over Prolog terms rather454

than being objects of a constraint domain. On the other hand, for a given size,455

exploring grounding to propositional formulas or answer-set programming seems456

worth exploring as a way to take advantage of today’s fast SAT-solvers.457

7 Related work458

The classic reference for lambda calculus is [1]. Various instances of typed lambda459

calculi are overviewed in [3].460

The combinatorics and asymptotic behavior of various classes of lambda461

terms are extensively studied in [5, 8]. Distribution and density properties of462

random lambda terms are described in [6].463

Generation of random simply-typed lambda terms and its applications to464

generating functional programs from type definitions is covered in [20].465

Several concepts of size have been used in the literature, partly to facilitate466

convergence of formal series in analytic combinatorics [21, 22].467

Asymptotic density properties of simple types (corresponding to tautologies468

in minimal logic) have been studied in [23] with the surprising result that most469

classical tautologies are also intuitionistic ones.470

In [4] a “type-directed” mechanism for the generation of random terms is471

introduced, resulting in more realistic (while not uniformly random) terms, used472

successfully in discovering some bugs in the Glasgow Haskell Compiler (GHC).473

Generators for closed simply-typed lambda terms, as well as their normal474

forms, expressed as functional programming algorithms, are given in [5], derived475

from combinatorial recurrences. However, they are significantly more complex476

than the ones described here in Prolog, and limited to terms up to size 10.477

In the unpublished draft [15] we have collected several lambda term gener-478

ation algorithms written in Prolog and covering mostly de Bruijn terms and a479

compressed de Bruijn representation. Among them, linear, affine linear terms as480

well as terms of bounded unary height and in binary lambda calculus encoding.481

In [15] type inference algorithms are also given for SK and X-combinator ex-482

pressions. A similar but about an order of magnitude slower program for type483

inference using de Bruijn notation is also given in [15], without however describ-484

ing the step-by-step derivation steps leading to it, as done in this paper.485

8 Conclusion486

We have derived several logic programs that have helped solve the fairly hard487

combinatorial counting and generation problem for simply-typed lambda terms488

4 orders of magnitude farther than previously published results.489

This has put at test two simple but effective program transformation tech-490

niques naturally available in logic programming languages: 1) interleaving gen-491

erators and testers by integrating them in the same predicate and 2) dropping492

arguments used in generators when used simply as counters of solutions, when493

their role can be kept implicit in the recursive structure of the program. Both494

have turned out to be effective for speeding up computations without chang-495

ing the semantics of their intended application. We have also managed (after a496

simple DCG translation) to work within in the minimalist framework of Horn497

Clauses with sound unification, showing that non-trivial combinatorics problems498

can be handled without any of Prolog’s impure features.499

Our techniques, combining unification of logic variables with Prolog’s back-500

tracking mechanism and DCG grammar notation, recommend logic program-501

ming as a convenient meta-language for the manipulation of various families of502

lambda terms and the study of their combinatorial and computational proper-503

ties.504

